Problem:
Score : 700 points

Problem Statement

Nuske has a grid with N rows and M columns of squares. The rows are numbered 1 through N from top to bottom, and the columns are numbered 1 through M from left to right.
Each square in the grid is painted in either blue or white. If S_{i,j} is 1, the square at the i-th row and j-th column is blue; if S_{i,j} is 0, the square is white.
For every pair of two blue square a and b, there is at most one path that starts from a, repeatedly proceeds to an adjacent (side by side) blue square and finally reaches b, without traversing the same square more than once.

Phantom Thnook, Nuske's eternal rival, gives Q queries to Nuske. The i-th query consists of four integers x_{i,1}, y_{i,1}, x_{i,2} and y_{i,2} and asks him the following: when the rectangular region of the grid bounded by (and including) the x_{i,1}-th row, x_{i,2}-th row, y_{i,1}-th column and y_{i,2}-th column is cut out, how many connected components consisting of blue squares there are in the region?

Process all the queries.

Constraints

1 ≤ N,M ≤ 2000

1 ≤ Q ≤ 200000

S_{i,j} is either 0 or 1.

S_{i,j} satisfies the condition explained in the statement.

1 ≤ x_{i,1} ≤ x_{i,2} ≤ N(1 ≤ i ≤ Q)

1 ≤ y_{i,1} ≤ y_{i,2} ≤ M(1 ≤ i ≤ Q)

Input

The input is given from Standard Input in the following format:

N M Q
S_{1,1}..S_{1,M}
:
S_{N,1}..S_{N,M}
x_{1,1} y_{i,1} x_{i,2} y_{i,2}
:
x_{Q,1} y_{Q,1} x_{Q,2} y_{Q,2}

Output

For each query, print the number of the connected components consisting of blue squares in the region.

Sample Input 1

3 4 4
1101
0110
1101
1 1 3 4
1 1 3 1
2 2 3 4
1 2 2 4

Sample Output 1

3
2
2
2

In the first query, the whole grid is specified. There are three components consisting of blue squares, and thus 3 should be printed.

In the second query, the region within the red frame is specified. There are two components consisting of blue squares, and thus 2 should be printed.
Note that squares that belong to the same component in the original grid may belong to different components.

Sample Input 2

5 5 6
11010
01110
10101
11101
01010
1 1 5 5
1 2 4 5
2 3 3 4
3 3 3 3
3 1 3 5
1 1 3 4

Sample Output 2

3
2
1
1
3
2

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program nuske_vs_phantom_thnook
  implicit none
  integer :: n, m, q, a(0:2000,0:2000), x1, y1, x2, y2, i, j, k
  integer :: x(0:2000,0:2000), y(0:2000,0:2000), z(0:2000,0:2000)
  character(2000) :: s
  a = 0
  x = 0
  y = 0
  z = 0
  read(*,*) n, m, q
  do i = 1, n
    read(*,*) s
    s = trim(adjustl(s))
    do j = 1, m
      if (s(j:j).eq."1") a(i,j) = 1
    end do
  end do
  do i = 1, n
    do j = 1, m
      x(i,j) = x(i,j-1)+x(i-1,j)-x(i-1,j-1)+a(i,j)
      y(i,j) = y(i,j-1)+y(i-1,j)-y(i-1,j-1)+a(i,j)*a(i,j-1)
      z(i,j) = z(i,j-1)+z(i-1,j)-z(i-1,j-1)+a(i,j)*a(i-1,j)
    end do
  end do
  do k = 1, q
    read(*,*) x1, y1, x2, y2
    write(*,'(i0)') f(x,x1,y1,x2,y2)-f(y,x1,y1+1,x2,y2)-f(z,x1+1,y1,x2,y2)
  end do
  stop
contains
  integer function f(x,i1,j1,i2,j2)
    implicit none
    integer, intent(in) :: x(0:2000,0:2000), i1, j1, i2, j2
    f = x(i2,j2)-x(i2,j1-1)-x(i1-1,j2)+x(i1-1,j1-1)
    return
  end function f
end program nuske_vs_phantom_thnook